XVII-th International Symposium on Electrical Apparatus and Technologies

SIELA 2012

PROCEEDINGS

Volume I

28-30 May 2012
Bourgas, Bulgaria

Union of Electronics, Electrical Engineering and Telecommunications (CEEC)
Technical Universities of Sofia and Varna
Ruse University „Angel Kantchev"
Federation of Scientific and Technical Unions
Centre of Informatics and Technical Sciences at Bourgas Free University
Houses of Science and Technology – Bourgas, Varna and Montana

XVII-th International Symposium on Electrical Apparatus and Technologies

SIELA 2012

PROCEEDINGS
Volume I

28-30 May 2012
Bourgas, Bulgaria

ISSN 1314-6297

COMPUTER MODELING OF RF MEMS INDUCTORS USING VHDL AMS LANGUAGE

Elissaveta GADJEVA*, Georgi VALKOV **

* Technical University of Sofia, Department of Electronics, 1156 Sofia, Bulgaria, E-mail: egadjeva@tu-sofia.bg

** Technical University of Sofia, Department of Electronics, 1156 Sofia, Bulgaria, E-mail: gvalkov@abv.bg

Abstract. On-chip inductors are important elements for the design of integrated RF circuits. A number of micromachining technologies, implemented in microelectromechanical systems (MEMS), are applied in RF applications. In the present paper, computer models are developed for RF MEMS inductors, using the standard VHDL-AMS language. They are simulated and verified in the Dolphin Integration SMASH simulation environment. Parameterized inductor macromodels are developed taking into account the frequency dependence of the series resistance due to the skin-effect.

Keywords: Behavioral computer models, equivalent circuits, RF MEMS inductors, VHDL-AMS language.

INTRODUCTION

The MEMS technology allows the production of micromachined inductors in which the parasitic capacitance and lossy substrate effects are alleviated. These inductors have enhanced *Q*-factor, increased selfresonant frequency, lower energy dissipation and lower phase noise in comparison to CMOS inductors [1-5]. With the development of instruments for Analog Behavioral Modeling (ABM) and simulation, it is possible to combine models defined in different standard languages and abstraction levels into a single project, in order to verify the behavior of the entire system. In the present paper, parameterized computer models for RF MEMS inductors are realized in the standard VHDL-AMS language. The models are simulated in the mixed-language, multi-domain environment provided by Dolphin SMASH [7].

RF MEMS INDUCTOR MODELS

Air suspended RF MEMS inductor model

The Π - RF physical planar inductor model shown in Fig. 1 [1, 5, 6] describes the performance of an air suspended RF MEMS inductor [1]. The model has been used extensively and has been proven to fit with *Y*- and *S*-parameter measurements of planar inductors. L_S is the low frequency inductance, C_S is the capacitance between the windings of the inductor, C_1 and C_2 are the capacitances in the oxide (or polyamide) layer between the coil and the silicon (or *GaAs*) substrate, C_{P1} and C_{P2} are the capacitances between the coil and the ground through the silicon substrate, and R_{P1} and R_{P2} represent the eddy current losses in the substrate, R_S is the series resistance of the coil [1].



Figure 1. Air suspended RF MEMS inductor.

The frequency dependence of R_S due to the skin-effect is represented by expression (1), where the value of f is in GHz:

(1) $$R_S(f) = A\sqrt{f}\ .$$

Simplified RF MEMS inductor model

In general R_{P1} and R_{P2} from Fig. 1 can be neglected and C_1 and C_{P1} are lumped together in one capacitance C_{P1}, the same applies to C_2 and C_{P2} [1], producing a simplified variant of the model, as shown in Fig. 2.



Figure 2. Simplified RF MEMS inductor model.

The series resistance R_S is assumed constant up to frequency f_0 and then increases as $\sqrt{f}$ to model the skin-effect [1]:

$$(2) \qquad R_S(f) = \begin{cases} R_{SO} & \text{for } f < f_o \\ R_{SO}\sqrt{\dfrac{f}{f_o}} & \text{for } f \geq f_o \end{cases}.$$

VHDL-AMS REALIZATION OF MEMS INDUCTOR MODELS

Air suspended RF MEMS inductor model

The VHDL-AMS code presented in Fig. 3 implements the model of air suspended RF MEMS inductor from Fig. 1, where the frequency dependence of R_S (1) is implemented as a function, as shown in Fig. 4.

```
library IEEE;
  use IEEE.electrical_systems.all;
  use IEEE.math_real.all;
entity inductor_mems_pi is
  generic (
        Cs  : capacitance:=  1.14e-15;
        Ls  : inductance :=  1.34e-9;
        C1  : capacitance:= 11.6e-15;
        C2  : capacitance:= 90.5e-15;
        Cp1 : capacitance:=  1.0e-15;
        Cp2 : capacitance:= 10.2e-15;
        Rp1 : resistance :=275.0;
        Rp2 : resistance :=332.0;
        A   : resistance :=  0.27
     );
  port(terminal n1, n2 : electrical);
end entity inductor_mems_pi;
architecture ideal of inductor_mems_pi is
  terminal n_s1, n_s2 : electrical;
  quantity U   across Ics through n2   to n1;
  quantity            Ils through n2   to n1;
  quantity Uc1 across Ic1 through n1   to n_s1;
  quantity Uc2 across Ic2 through n2   to n_s2;
  quantity Up1 across Ip1 through n_s1 to electrical_ref;
  quantity Up2 across Ip2 through n_s2 to electrical_ref;
begin
    U   == Ls *Ils'dot + Ils*Rs(frequency);
    Ics == Cs *U  'dot;
    Ic1 == C1 *Uc1'dot;
    Ic2 == C2 *Uc2'dot;
    Ip1 == Cp1*Up1'dot + Up1/Rp1;
    Ip2 == Cp2*Up2'dot + Up2/Rp2;
end architecture;
```

Figure 3. VHDL-AMS code of the air suspended MEMS inductor model.

```
  function Rs(freq: real) return real is begin
    return A*sqrt(freq*1.0e-9);
  end function;
```

Figure 4. VHDL-AMS implementation of the frequency dependence of R_S due to skin effect.

To verify the model, four sample inductors defined in Table 1 are simulated and compared to measurement data from [4]. The simulation results are shown in Fig. 5.

Table 1. Lumped-element parameters of fabricated inductors

#	Ls (nH)	A	Cs (fF)	C_1 (fF)	C_2 (fF)	C_{P1} (fF)	C_{P2} (fF)	R_{P1} (Ω)	R_{P2} (Ω)
L_1	1.34	0.27	1.14	11.6	90.5	1	10.2	275	332
L_2	1.77	0.39	0.82	20.2	92.2	27.6	21.2	193	222
L_3	2.52	0.64	0	25.8	160	34.4	20.3	230	167
L_4	3.90	0.99	0	40.0	184	43.8	16.7	199	148



a) b)

Figure 5. Simulated inductance (a) and Q-factor (b) for air suspended MEMS inductor.

A comparison between measured and simulated data for Q_2 is shown in Fig. 6. The average relative error is 2.6%.



Figure 6. Comparison between measured and simulated data for Q_2.

Simplified RF MEMS inductor model

The VHDL-AMS code presented in Fig. 7 implements the simplified model of MEMS inductor from Fig. 2, where the frequency dependence of R_S (2) is implemented as a function, as shown in Fig. 8.

```
library IEEE;
  use IEEE.electrical_systems.all;
  use IEEE.math_real.all;
entity inductor_mems_simple is
  generic (
      Ls  : inductance :=  5.0e-9;
      Cs  : capacitance:=  9.0e-15;
      Cp1 : capacitance:= 75.0e-15;
      Cp2 : capacitance:= 75.0e-15;
      Rso : resistance :=  6.3;
      fo  : real       :=  2.0e+9
  );
  port(terminal n1, n2 : electrical);
end entity inductor_mems_simple;
architecture ideal of inductor_mems_simple is
  quantity U   across Ics through n2 to n1;
  quantity            Ils through n2 to n1;
  quantity Up1 across Ip1 through n1 to electrical_ref;
  quantity Up2 across Ip2 through n2 to electrical_ref;
begin
  U   == Ls *Ils'dot + Ils*Rs(frequency);
  Ics == Cs *U  'dot;
  Ip1 == Cp1*Up1'dot;
  Ip2 == Cp2*Up2'dot;
end architecture;
```

Figure 7. VHDL-AMS code of the simplified RF MEMS inductor model.

```
  function Rs(freq: real) return real is
  begin
      if(freq < fo) then return Rso;
      else return Rso*sqrt(freq/fo); end if;
  end function;
```

Figure 8. VHDL-AMS implementation of the frequency dependence of R_S due to skin effect.

Fig. 9 presents the simulated quality factor Q for $L_S = 5$ nH, $R_{SO} = 6.3\ \Omega$, $C_{P1} = C_{P2} = 75$ fF, $C_S = 9$ fF. The effect of the series resistance on the Q is shown in Fig. 10. The effect of the substrate capacitance on the Q is shown in Fig. 11.



Figure 9. Simulated Q for the simplified RF MEMS inductor.



Figure 10. Effect of the series resistance on Q for simplified RF MEMS inductor.



Figure 11. Effect of the substrate capacitance on Q for simplified RF MEMS inductor.

A comparison between measured and simulated data for Q is shown in Fig. 12. The average relative error is 3.5%.



Figure 12. Comparison between measured and simulated data for Q.

CONCLUSIONS

Parameterized behavioral computer models for RF MEMS inductors have been developed using the standard VHDL-AMS language. The frequency dependence of the series resistance due to the skin-effect is taken into account. The simulation results are in agreement with the measurement data. The average relative error is 2.6% for the air suspended RF MEMS inductor model and 3.5% for the simplified RF MEMS model.

ACKNOWLEDGEMENT

The investigations are supported by the project №122PD0026-03.

REFERENCES

[1] G. Rebeiz. *RF MEMS: Theory, design, and technology*. John Wiley & Sons, Inc., Hoboken, New Jersey 2003.

[2] S. Seok, C. Nam, W. Choi and K. Chun. A high performance solenoid-type MEMS inductor. Journal of Semiconductor Technology and Science, vol. 1, n. 3, June 2001.

[3] T. Merkin, S. Jung, S. Tjuatja, Y. Joo, D. S. Park and J. B. Lee. An ultra-wideband low noise amplifier with air-suspended RF MEMS inductors. The 2006 IEEE 2006 International conference on ultra-wideband, 2006, pp. 459-464.

[4] J. Yoon, Y. Choi, B. Kim, T. Eo and E. Yoon. CMOS compatible surface-micromachined suspended-spiral inductors for multi-GHz silicon RF ICs. IEEE electron device letters, vol. 23, n. 10, Oct. 2002, pp. 591-593.

[5] C. Yue, C. Ryu, J. Lau, T. Lee, and S. Wong. A physical model for planar spiral inductors on silicon. Proc. IEEE Int. electron devices meeting tech. dig., San Francisco, Dec. 1996, pp. 155-158.

[6] E. Gadjeva, V. Durev and M. Hristov. Matlab - Modelling, programming and simulations. Chapter 14: Analysis, model parameter extraction and optimization of planar inductors using MATLAB, Published by SCIYO, Copyright © 2010 Sciyo, ISBN 978-953-307-125-1, www.sciyo.com, http://www.intechweb.org/ .

[7] Dolphin Integration SMASH overview, http://www.dolphin.fr/medal/smash/smash_overview.php .